
Abstract. Constraint Handling Rules (CHR) is a concurrent, commit-
ted-choice, rule-based language, rewriting constraints in relational syn-
tax. Frame-Logic is an extension of classical predicate logic which ac-
counts in a declarative way for many features of object-orientation. This
exploratory paper gives a concise CHR implementation of Frame-Logic’s
core features, including object-oriented constraint syntax, type-checking,
and interaction of Frame-Logic deduction with non-monotonic overriding
by inheritance.

1 Introduction

Constraint Handling Rules (CHR) [5] is a concurrent, committed-choice, rule-
based language which was originally developed for rapid prototyping and for
writing constraint solvers.

Frame-Logic (F-Logic) [7] is an extension of classical predicate logic, which
accounts in a declarative way for many features of object-orientation: Object-
oriented syntax breaks the so-called impedance mismatch between (relational)
logic and object-oriented programming. Object methods are encapsulated and
not scattered around relations. Overriding of inherited definitions, common in
standard programming languages like Java or C++, is modelled by allowing
inheritance to be non-monotonic. F-Logic allows intricate interaction between
rule execution, i.e., F-Logic deduction, and inheritance [13]. Our motivation
for using F-Logic is that object-orientation is described by F-Logic as relational
programming is by classical predicate logic. As the classical declarative reading of
a CHR program is given in classical predicate logic, an object-oriented extension
of CHR should be described in the object-oriented F-Logic.

F-Logic is orthogonal to other extensions of classical predicate logic like (non-
monotonic) transaction logic [3] and (syntactically higher-level) HiLog [4]. All
three extensions are implemented in the elaborate FLORA-2 system [14,6], which
is based on general logic programming and uses XSB’s [11] implementation of
the three-valued well-founded semantics. However, a thorough background both
in theory and implementation is a prerequisite for using FLORA-2, so it would
be much nicer to express parts of F-Logic in terms of a programmable language
such as CHR. In this exploratory paper we show that our CHR implementation
accounts for the basic F-Logic features including object-oriented syntax, F-Logic
deduction, inheritance, and type-checking.

Beside the trend to use CHR as general purpose programming language, we
see two long-term benefits of investigating CHR and object orientation (cf. [9]):
First, object-oriented syntax and non-monotonic inheritance semantics is crucial
to use CHR as reasoning engine in mainstream software engineering. Second,
understanding object-orientation in terms of a rule-based language should make
it easier to apply mainstream object-oriented software engineering principles on
a to be devised object-oriented CHR dialect.

Our paper is organised as follows: After introducing F-Logic in Section 2, we
describe our implementation in detail in Section 3, and conclude.

Complete sources and examples for our implementation (for SICStus Pro-
log [10]) are available [1].

2 F-Logic

We give an introduction to F-Logic: An example illustrates important aspects of
F-Logic (Section 2.1). We describe the syntax of F-Logic expressions (Section 2.2)
and explain the semantics of inference both by rules and by inheritance and their
interaction (Section 2.3).

In order to exhibit the core concepts of F-Logic we focus on set-valued method
definitions only (as proposed by Kifer [6]).

2.1 Motivating Example

We give an overview of F-Logic by means of an example. An F-Logic program
contains F-facts, is-a atoms, and object atoms. The relation between objects is
defined by is-a atoms:

kaylee : goodguy, simon : goodguy, jayne : badguy, mal : person,
goodguy :: person, badguy :: person.

a :: b defines a subclass relationship, while a : b defines a class membership. For
example, kaylee is a class member (instance) of goodguy, which is a subclass of
person.

An object atom o[p] defines a method expression p for an object o:

person[attitude •→→ friendly],
badguy[attitude •→→ aggressive],
kaylee[attitude→→ shiny].

The data expression b →→ c is non-inheritable, while b •→→ c is inheritable, i.e.,
influencing additionally its subclasses and instances. Methods are always set-
valued, so multiple expressions are possible with different values for the same
object and method. For example, in addition to kaylee[attitude →→ shiny] the
expression kaylee[attitude→→ aggressive] is allowed.

Objects and classes are not mutually exclusive, e.g., person has class prop-
erties (i.e., it has instances and subclasses) and object properties (i.e., it has
method expressions), while kaylee has only object properties.

F-Logic provides implicit inference by inheritance. If not overwritten, in-
heritable expressions are passed to its subclasses also as inheritable, but to its
instances as non-inheritable: goodguy is a subclass of person, so the expres-
sion goodguy[attitude •→→ friendly] is inherited. The objects simon and kaylee
are instances of goodguy, so simon inherits attitude →→ friendly, but kaylee
overrides attitude to be shiny.

Information is inferred explicitly by F-rules, which may be part of an F-Logic
program:

X[likes→→ X]←− X : badguy,
X[likes→→ Y ]←− X : goodguy and Y : goodguy.

If the expression in the body is satisfied for some variable assignment, the atom
in the head is also considered true. The examples tell that every goodguy likes
each goodguy, while every badguy only likes himself, so the following F-facts are
inferred:

jayne[likes→→ jayne],
kaylee[likes→→ simon], kaylee[likes→→ kaylee],
simon[likes→→ simon], simon[likes→→ kaylee].

F-Logic supports F-queries, i.e., for an expression with (possibly) variables,
e.g., we ask for those variable assignments that satisfy

?−X[likes→→ kaylee] and X[attitude→→ Y ].

and get the answer

(X = kaylee and Y = shiny) or (X = simon and Y = friendly)

2.2 Syntax

Since the original reference [7], the syntax of F-Logic was modernised and sim-
plified. We distinguish set/single-valued F-expressions by cardinality constraints
(as proposed in [6]), and use only one type of arrow: The two-headed arrow fits
Prolog’s syntax nicely.

Following the usual convention, variables start with uppercase letters and
function symbols start with lowercase letters. An id-term is a first-order term
composed of function symbols and variables, A logical object id (short: oid), that
is used to identify an object, is a ground (i.e., variable-free) id-term.

F-atoms are the basic elements of F-Logic:

is-a atom Defines a relation between two id-terms: A : B indicates that A is a
class member (instance) of B, while A :: B means that A is a subclass of B.
Example: simon : goodguy, goodguy :: person, X : object.

object atom O[E] defines a method expression E for an id-term O. E is either
a data expression (M •→→ V , M →→ V ) or a signature expression (M •⇒⇒ V ,
M ⇒⇒ V ), that each can be inheritable or non-inheritable.
Example: person[likes •⇒⇒ person], kaylee[attitude→→ shiny].

F-formulas are expressions constructed of F-atoms, the usual logical connec-
tives negation, ∧, ∨, and quantifiers. Special F-formulas are:

F-query Expression consisting of F-atoms and the logical connectives ∧ and ∨.
F-rule H ←− B for a universally quantified expression ∀(¬B ∨ H) (H is a

conjunction of F-atoms, and B is an F-query).
F-fact Special case of F-rule: F ←− true.

An F-molecule is a compact notation for a conjunction of F-atoms, e.g.,
a : c[m •→→ [b1, b2],m •⇒⇒ c] means a : c∧a[m •→→ b1]∧a[m •→→ b2]∧a[m •⇒⇒ c].

We call F-facts, F-rules and F-queries consisting of F-molecules instead of
F-atoms molecular.

2.3 Semantics: Inheritance and Rule Application

An F-Logic program (short: F-program) P is a conjunction of F-rules. The corre-
sponding Herbrand model M is the minimal set of ground F-atoms, that satisfies
the F-Logic program:

M1 ∧ ... ∧Mn |= P for M = {M1, ...,Mn}

Obviously, every F-fact from the F-program is part of the model. Further,
for every variable assignment, for which the body of an F-rule is satisfied by the
model, the head is part of the model, e.g.:

P = kaylee[likes→→ simon] ∧ ∀(X[likes→→ Y ]←− Y [likes→→ X])
M = {kaylee[likes→→ simon], simon[likes→→ kaylee]}

We distinguish structural inheritance for is-a and inheritable signature atoms,
and behavioural inheritance for inheritable data atoms.

The model for monotonic, structural inheritance satisfies:

A :: B ∈M ∧ B :: C ∈M =⇒ A :: C ∈M
A : B ∈M ∧ B :: C ∈M =⇒ A : C ∈M

A :: B ∈M ∧ B[C •⇒⇒ D] ∈M =⇒ A[C •⇒⇒ D] ∈M
A : B ∈M ∧ B[C •⇒⇒ D] ∈M =⇒ A[C ⇒⇒ D] ∈M

Behavioural inheritance is non-monotonic, i.e., an additional F-atom can lead
to the removal of another F-atom in the model. Model-theoretic semantics is
rather complicated, informally, we expect

A :: B ∈M ∧ B[C •→→ D] ∈M =⇒ A[C •→→ D] ∈M
A : B ∈M ∧ B[C •→→ D] ∈M =⇒ A[C →→ D] ∈M

unless the inheritance is overridden by an F-atom in an intermediate or the
destination object. Consider this F-program:

kaylee : goodguy ∧ goodguy :: person ∧ person[attitude •→→ friendly]

Hence, kaylee : person is inherited structurally, while two F-atoms are added to
the model by behavioural inheritance:

goodguy[attitude •→→ friendly], kaylee[attitude→→ friendly]

If goodguy[attitude •→→ shiny] is added, the last two F-atoms are removed
from the model. Inheritance is blocked, as for goodguy the destination object
is overridden, while for kaylee, an intermediate object is overridden. Instead
kaylee inherits from goodguy: kaylee[attitude→→ shiny].

The interaction of F-rules and non-monotonic inheritance leads to several
subtle problems (c.f. [6]), e.g.,

a :: b ∧ b :: c ∧ c[m •→→ v1] ∧ b[m •→→ v2]←− a[m •→→ v1]

Since no overriding occurs, a[m •→→ v1] should be inherited, therefore b[m •→→ v2]
should be part of the model. But this method should inhibit the first inheritance
in favour of a[m •→→ v2], and this would undermine the F-rule application.

The original solution from Kifer [7] was not model-theoretic and was prob-
lematic in several other respects as well. A satisfactory and completely model-
theoretic solution was proposed by Yang [12,13].

3 CHR Implementation of F-Logic

The purpose of our F-Logic handler for CHR is making important parts of F-
Logic executable, but keeping the implementation concise and comprehensible.
To this end, our implementation covers (up to now), only a subset of the very
powerful F-Logic formalism.

For an F-Logic program, which is a conjunction of F-facts and F-rules, we
try to build a suitable Herbrand model by forward propagation. We expand the
set of F-facts by repeatedly applying F-rules and inheritance and hence inferring
new F-facts, until a stable state is reached. This set of F-facts is the model for
this F-Logic program, i.e., the minimal set of F-facts that satisfies the program.
Finally an F-query can be answered using this model.

While the current forward chaining approach facilitates interaction of deduc-
tion and inheritance, we might need a more sophisticated goal-driven mechanism
for efficiency reasons in the future.

In Section 3.1 we show our representation for F-atoms and F-molecules and
their decomposition. We illustrate deduction by F-rules (Section 3.2) and the
similar mechanism for answering F-queries. We describe inference by structural
and behavioural inheritance (Section 3.3). Optional check for type correctness
w.r.t. signature expressions is explained (Section 3.4). Finally the effects of in-
teraction between F-rules and inheritance are discussed with several examples
(Section 3.5).

Complete sources and examples for our implementation (for SICStus Pro-
log [10]) are available [1].

3.1 F-Logic Facts in CHR

For F-facts we allow only ground terms. For convenience, F-molecules are sup-
ported and decomposed to F-atoms, so we describe F-atoms at first.

Our F-Logic syntax for CHR is intended to be visually as close as possible to
the F-Logic notation proposed by Kifer [7]. The three categories of F-atoms are
represented as CHR constraints: a..b for class membership, a::b for a subclass
relation, and a~(b) for an object atom; data expressions are encoded as m->>o
and m*->>o, signature expressions as m=>>o and m*=>>o (cf. Table 1), this is a
total of six F-atom types. We call the components of F-atoms arguments, e.g.,
a, b, m and o in the last F-atoms. We define Prolog infix operators with adapted
precedences.

Table 1. F-atoms in CHR syntax

F-Logic Syntax CHR-Syntax

a : b a..b

a :: b a::b

a[m →→ x] a~(b->>c)

a[m •→→ x] a~(b*->>c)

a[m ⇒⇒ x] a~(b=>>c)

a[m •⇒⇒ x] a~(b*=>>c)

The handler removes duplicate F-facts to prevent infinite computations by
cyclic rules, e.g., A : B ←− A : B.

F-molecules are decomposed into F-atoms (cf. Table 2), e.g., the molecule
a..c~[m*->>[[b1,b2]..c], m*=>>c] is converted to the atoms a..c, b1..c,
b2..c, a~(m*->>b1), a~(m*->>b2), and a~(m*=>>c).

Table 2. Decomposition of F-molecules into F-atoms

F-molecule F-atoms

a~[b,c] a~(b), a~(c)

[a,b]..c a..c, b..c

[a,b]::c a::c, b::c

a..b~(c) a..b, a~(c)

a::b~(c) a::b, a~(c)

a~(b->>[c1,c2]) a~(b->>c1), a~(b->>c2)

a~(b->>c~(d)) a~(b->>c), c~(d)

a~(b->>c..d) a~(b->>c), c..d

a~(b->>c::d) a~(b->>c), c::d

These are some of the CHR rules dealing with decomposition:

O~[H|T] <=> O~H, O~T.

O~(M->>[H|T]) <=> O~(M->>H), O~(M->>T).
O~(M->>V..C) <=> O~(M->>V), V..C.

The first CHR rule handles objects with multiple expressions, the second meth-
ods with multiple values, and the last decomposes nested expressions.

3.2 F-Rule Application and F-Queries

F-Rule Application An F-Rule R is represented as H <-- B with head H being
an F-atom and body B being an expression with conjunctions and disjunctions
of F-atoms, the arguments of F-atoms may be variables or ground terms:

R := H <-- B
H := F-atom
B := F-atom | B or B | B and B

At first, the F-rules are transformed, s.t. each body is a conjunction. Then,
for each F-rule, we search for all variable instantiations that satisfy its body,
i.e., that the corresponding F-facts do exist. At last, we apply each satisfying
instantiation on the corresponding head and propagate it. The deduced F-facts
however must be ground terms.

First, the handler converts expressions in the body to disjunctive normal
form (DNF), and splits the F-rule for every part of the disjunction, so that each
body is a conjunction of F-atoms:

Head <-- B and (C or D) <=> Head <-- (B and C) or (B and D).
Head <-- (B or C) and D <=> Head <-- (B and D) or (C and D).
Head <-- B or C <=> Head <-- B, Head <-- C.

For storing variable bindings, we introduce a partially satisfied rule: the CHR
constraint frule/3 for F-rules with equality restrictions.

Let H, B1,...,Bn be F-atoms, X1,...,Xm be ground terms or variables and
Y1,...,Ym be ground terms, then

frule(H<--[B1,...,Bn], [X1,...,Xm],[Y1,...,Ym])

is defined by

H ←− B1 and ... and Bn and X1 = Y1 and ... and Xm = Ym

For every F-rule, we convert the conjunction in the body to a list and prop-
agate an frule without equality restrictions.

For example, the F-rule H <-- B and (C or D) is pre-processed by the fol-
lowing steps (let H, B, C and D be F-atoms):

– H <-- B and (C or D).
– H <-- (B and C) or (B and D).
– H <-- B and C, H <-- B and D.
– frule(H<--[B,C], [],[]), frule(H<--[B,D], [],[]).

Second, we search matching F-facts for every F-atom in the frule body and
store the corresponding variable bindings. One CHR rule deals with each of the
six types of F-atoms, e.g., consider this code fragment:

frule(Head <-- [Q1..Q2 |BR], X,V), P1..P2
==> X2=[Q1,Q2|X], V2=[P1,P2|V], subsumes_chk(X2,V2) |

frule(Head <-- BR, X2,V2).
frule(Head <-- [Q1~(Q2*->>Q3)|BR], X,V), P1~(P2*->>P3)
==> X2=[Q1,Q2,Q3|X], V2=[P1,P2,P3|V], subsumes_chk(X2,V2) |

frule(Head <-- BR, X2,V2).

Matching of the first piece of the F-rule body and an F-fact is possible if the
F-atoms are of the same type and its arguments can be unified taking into
account the previous variable bindings. We test for unification with the Prolog
predicate subsumes chk1. In the following examples, frule R and F-fact F can
not interact as subsumes chk fails (with H replaced by a proper head):

– Ground term should equal different ground term:
R=frule(H<--[X..goodguy], [],[]), F=jayne..badguy.
Here subsumes_chk([X,goodguy],[jayne,badguy]) fails.

– Same variable should equal different ground terms:
R=frule(H<--[X~(likes->>X)], [],[]), F=kaylee~(likes->>simon).
Here subsumes_chk([X,likes,X],[kaylee,likes,simon]) fails.

– Previously bound variable should equal different ground term:
R=frule(H<--[X..goodguy], [X],[mal]), F=kaylee..goodguy.
Here subsumes_chk([X,goodguy,X],[kaylee,goodguy,mal]) fails.

If the test for matching succeeds, we propagate a new frule-constraint with
the first part of the body removed, but equality restrictions2 added.

We add the elements, i.e., object, method, and return value, of the F-rule’s
body piece to the list in the second argument of frule. In parallel, we add the
elements of the F-atom, that matches the body, to the list in the third argument
of frule. As the list in the third argument is ground, subsumes chk can check
for unification.

We use CHR propagation instead of CHR simplification rules to get all vari-
able bindings from different matching F-facts.

Compared to its ancestor, the list of F-atoms in the body of an frule is
decreased by one. For an empty body the frule is satisfied solely by the equality
restrictions, so finally we apply them on the head and propagate it as a new F-
fact:
1 subsumes chk(A,B) tests if A subsumes B, i.e., whether an instantiation of A exists,

so that A=B, e.g., it succeeds for A=[X,X,Y,c], B=[a,a,c,c], but fails for A=a, B=X
or A=a, B=b or A=[X,X], B=[a,b]. If B is a ground term like in our case, a test for
subsumption is also a test for unification.

2 Equality restrictions are guaranteed contradiction-free and either of the useful form
X = a or the trivial form a = a.

frule(Head <-- [], [X1|XR],[V1|VR]) <=>
instantiate(X1=V1,Head,Head2) | frule(Head2 <-- [], XR,VR).

frule(A~B <-- [], [],[]) <=> A~B.

The user-defined predicate instantiate(X=V, Term, Term2) replaces all
occurrences of X in Term by V and copies the result to Term2. Note that the
variables in the term are instantiated manually by digging into the term struc-
ture. We do not use build-in unification to leave the variables in the F-rules
untouched.

F-Queries An F-Query ? Q is a shortcut for the pseudo F-rule query <-- Q.
The same CHR rules are used for F-queries, except for the third and last step,
i.e., dealing with empty bodies:

frule(query <-- [], X,V) <=> frule(query([]) <-- [], X,V).
frule(query(L) <-- [], [X1|XR],[V1|VR]) <=> var(X1) |

frule(query([X1=V1|L]) <-- [], XR,VR).
frule(query(L) <-- [], [X1|XR],[_|VR]) <=> ground(X1) |

frule(query(L) <-- [], XR,VR).
frule(query(L) <-- [], [],[]) <=>

remove_duplicates(L,L2), reverse(L2,L3) | result(L3).

Ignoring trivial equality constraints (third CHR rule), we collect the remain-
ing equalities (second CHR rule), remove duplicates, and store the final list in
the constraint result/1 (fourth CHR rule).

A separate result is returned for each valid set of variable instantiations for
each disjunctive part of the F-query. A single result contains one value for each
variable. Hence the answer is in disjunctive normal form, e.g.: result([X=a]),
result([X=b,Y=c] corresponds to X = a or (X = b and Y = c).

Consequently for an F-query without solution no result is returned. For a
valid F-query without variables we return the unrestricted result([]).

kaylee..goodguy, simon..goodguy,
X~(likes->>Y) <-- X..goodguy and Y..goodguy,
? Who~(Meth->>kaylee).

The following CHR constraints are derived from this F-Logic program:

kaylee~(likes->>simon), kaylee~(likes->>kaylee),
simon~(likes->>simon), simon~(likes->>kaylee),
result([Who=kaylee, Meth=likes]),
result([Who=simon, Meth=likes]).

Therefore the answer to the F-query is:

(Who = kaylee and Meth = likes) or (Who = simon and Meth = likes)



10 Martin Kaeser and Marc Meister

3.3 Inheritance

While structural inheritance, responsible for is-a atoms and inheritable signature
expressions, takes place in any case, behavioural inheritance for inheritable data
expressions may be blocked by overriding.

On the one hand, structural inheritance is implemented by four simple CHR
rules:

O..C1, C1::C2 ==> O..C2.
C1::C2, C2::C3 ==> C1::C3.
O..C, C~(M*=>>V) ==> O~(M=>>V).
S::C, C~(M*=>>V) ==> S~(M*=>>V).

On the other hand, behavioural inheritance is more complicated: A data
expression must not be inherited immediately, but must allow F-facts, that are
defined afterwards or are derived by F-rules, to override the method. So the user
must explicitly trigger inheritance by completing an F-Logic program with the
constraint inheritance.

We create a candidate constraint for every potentially inheriting method.
Afterwards we remove candidates that originate from inherited methods or are
overridden by a more specific candidate or a method definition. The trigger then
selects one candidate for actual inheritance.

For inheritance to both subclasses and members, we create a candidate with
type, destination and source object, and method identifier:

O..C, C~(M*->>_) ==> candidate(member,O,C,M).
S::C, C~(M*->>_) ==> candidate(class, S,C,M).

Note that several identical candidates are created for methods with multiple
values, however these duplicates are removed by a CHR rule.

We do not allow inherited methods to be inherited further. This important
point is explained at the end of the section. We identify these troublesome meth-
ods c~(m*->>_) by an existing inherit(class,c,_,m) (inherit has the same
structure as candidate, but indicates certain inheritance; see below). We remove
the corresponding candidate:

inherit(class,C,_,M) \ candidate(_,_,C,M) <=> true.

Two candidates are called competing, if they are identical except for the
source object. For two competing candidates the less specific, i.e., whose source
object is a superclass of the other, is removed:

C1::C2, candidate(T,O,C1,M) \ candidate(T,O,C2,M) <=> true.

If the source objects of two competing candidates are incomparable, we might
have multiple inheritance, which is indicated by a warning. Candidates overrid-
den by an appropriate method definition of the destination object are removed:

O~(M->>_) \ candidate(member,O,_,M) <=> true.
O~(M*->>_) \ candidate(class, O,_,M) <=> true.

The trigger inheritance selects a remaining candidate for actual inheri-
tance and marks it with inherit:

inheritance, candidate(T,O,C,M) ==> inherit(T,O,C,M).

The constraint inherit has the effect, that the complete method is inher-
ited in any case, probably consisting of multiple expressions for multiple values.
Further competing candidates are overridden by newly inherited method expres-
sions. Therefore inheritance from competing candidates selects one “winner”
non-deterministically. To inhibit the selection of multiple competing candidates
we use CHR refined semantics and place the previous CHR rule at the end of the
inheritance block. The CHR rules for actually doing the inheritance are placed
before:

inherit(member,O,C,M), C~(M*->>V) ==> O~(M->>V).
inherit(class, S,C,M), C~(M*->>V) ==> S~(M*->>V).

We complete the section with an example:

mal..goodguy, mal..badguy, [goodguy,badguy]::person,
badguy~(attitude*->>aggressive),person~(attitude*->>friendly).

We clearly expect attitude to be inherited from person to goodguy, and from
badguy to mal. What will happen in our handler?

mal..person is inherited structurally, and the following candidates are cre-
ated:

candidate(member,mal,badguy,attitude),
candidate(class,goodguy,person,attitude),
candidate(member,mal,person,attitude).

The attitude for mal could potentially be inherited from badguy or person, but
because of being not so specific, the candidate for person is removed. Depending
on the sequencing of the F-facts, inheritance from person to goodguy may occur
first, then we get these constraints:

candidate(member,mal,badguy,attitude),
inherit(class,goodguy,person,attitude),
goodguy~(attitude*->>friendly),
candidate(member,mal,goodguy,attitude).

Now mal could potentially inherit from goodguy or badguy. But inheriting from
goodguy is in fact inheriting indirectly from person, which was rejected in favour
of badguy. Because of inheritance for is-a atoms, inheriting previously inherited
methods is never necessary. Therefore we do not permit it, we detect this by
means of inherit and remove candidate(member,mal,goodguy,attitude).
Finally we have:

goodguy~(attitude*->>friendly), mal~(attitude->>aggressive).

If we explicitly define goodguy~(attitude*->>friendly), however, we get
a multiple inheritance conflict and either friendly or aggressive is inherited
for mal’s attitude.

3.4 Type Checking

For every pair of corresponding data and signature expression (O~(M->>V),
O~(M=>>Type) and O~(M*->>V), O~(M*=>>Type), respectively), we expect the
type restriction V..Type to be fulfilled for type correctness. If this is not the
case, the program aborts with an error message.

To allow the membership relation after data and signature expression, type
checking should not occur immediately, so the user has to trigger type checking
with typecheck if desired. An arbitrary number (including zero) of signature
expressions is allowed per data expression, for type correctness every type re-
striction must be fulfilled.

Type checking is implemented by four CHR rules:

O~(M->>V), O~(M=>>Type) ==> expect(V..Type).
O~(M*->>V), O~(M*=>>Type) ==> expect(V..Type).
V..Type \ expect(V..Type) <=> true.
typecheck,expect(_) <=> write(’error: typecheck failed’),fail.

The following example is type correct:

friendly..normal, aggressive..bad,
jayne~[attitude->>friendly, attitude=>>normal],
badguy~[attitude*->>aggressive, attitude*=>>bad].

But if we add jayne..badguy, the signature jayne~(attitude=>>bad) is in-
herited structurally, therefore additionally friendly..bad is expected and type
checking fails.

3.5 Interaction between F-Rule Application and Inheritance

The interaction of inheritance and deduction leads to several anomalies. We
implement Kifer’s original fixpoint computation scheme for inheritance [7] and
discuss the behaviour of our handler w.r.t. three typical problematic examples
from Yang [13]. On the left of the figures we illustrate the object hierarchy, on
the right we show the F-Logic program. An arrow from node c1 to another node
c2 indicates that c1 is a subclass of c2.

In Figure 1 our handler inherits (m*->>a) from c2 to c1. Therefore through
F-rule application c2~(m*->>b) is inferred, which leads to further inheritance.
Altogether the following F-facts are inferred:

c1~(m*->>a), c2~(m*->>b), c1~(m*->>b).

c1

c2[m •→→ a]

c1::c2,

c2~(m*->>a),

c2~(m*->>b) <-- c1~(m*->>a).

Fig. 1. Superclass changed by rule

c1

c2[m •→→ a]

c1::c2,

c2~(m*->>a),

c1~(m*->>b) <-- c1~(m*->>a).

Fig. 2. Subclass changed by rule

Our implementation is correct for Yang’s first problem.
In Figure 2 only the head of the F-rule is different. Our handler again inherits

the method expression from c2 to c1, so the F-rule is applied and c1~(m*->>b)
is propagated. Therefore c1 has an inherited and a deduced method expression:

c1~(m*->>a), c1~(m*->>b).

The answer of our implementation differs from Yang’s, that interprets a deduced
expression as equivalent to a local definition, so it should override the method
and inhibit the previous inheritance, and therefore undermine the reason for
the preceding F-rule application. The only reasonable solution would be leaving
both F-facts undefined.

c1

c2[m •→→ a]

c3[m •→→ b]

c1::c2, c3::c2,

c2~(m*->>a),

c3~(m*->>b),

c1::c3 <-- c1~(m*->>a).

Fig. 3. Hierarchy changed by rule

In Figure 3 our handler inherits the method expression from c2 to c1, so
by F-rule application, c1::c3 is added and c3 becomes more specific than c2.
Nevertheless inheritance from c3 to c1 is not possible, because it is blocked by
c1~(m*->>a). Therefore we infer:

c1~(m*->>a), c1::c3.

Following Yang’s interpretation, however, the method expression for c1 should
be inherited from the more specific c3 instead of c2 and therefore undermine
the reason for the preceding F-rule application. Consequently all three F-facts
should be left undefined:

c1~(m*->>a), c1::c3, c1~(m*->>b).

As mentioned in the last two examples, our current handler’s fixpoint com-
putation does not deal with all interactions between inheritance and F-rules in a
reasonable way. As Yang’s suggested implementation of inheritance [13] relies on
general logic programming, a CHR implementation to handle it, will probably
be more difficult and left to future work.

4 Conclusion

We follow the trend to use CHR as a general purpose programming language. We
implemented the main features of F-Logic in CHR for refined semantics, includ-
ing object-oriented syntax, F-Logic deduction, inheritance, and type-checking.
Compared to the available FLORA-2 system, which uses general logic program-
ming and well-founded semantics, our CHR implementation of (parts of) F-Logic
is concise. Complete sources and examples for our implementation (for SICStus
Prolog [10]) are available [1].

Implementing F-Logic in CHR is a first step towards a hybrid object-oriented
rule-based constraint language. Currently, we investigate how the intricate inter-
action of F-rule deduction and (non-monotonic) inheritance can be implemented
in CHR. A full F-Logic implementation in CHR should be simpler and thus eas-
ier to use than the currently available FLORA-2 engine. Considering all (F-Logic
related) aspects of FLORA-2, including its performance scalability, our concise
and purely data driven forward chaining approach might need to be revised.

